
Сейчас, когда на рынке можно найти различные тензорезистивные преобразователи давления, как 

отечественного, так и импортного производства, выпуск различных датчиков давления, на их основе, уже 
не является такой большой проблемой как ранее. Пользуясь этим обстоятельством различные творческие 
коллективы предлагают свои услуги в этой области. И, не смотря на то, что их продукция порой не имеет 
сертификата соответствия, она находит спрос, благодаря своей низкой стоимости. Снижение стоимости 
обычно достигается тем, что покупаются самые дешёвые, некомпенсированные, тензорезистивные 
преобразователи давления, имеющие большую зависимость выходного сигнала от температуры 
кристалла. Приводимый в статье материал позволит, на основе некомпенсированных 
тензопреобразователей, выпускать качественные датчики давления, которые уже будет не стыдно 
сертифицировать в соответствующих государственных организациях.  

 
КОМПЕНСАЦИЯ ТЕМПЕРАТУРНОГО  ДРЕЙФА НУЛЯ 
ИНТЕГРАЛЬНОГО ТЕНЗОРЕЗИСТИВНОГО ДАТЧИКА 

ДАВЛЕНИЯ 
 
Первые сообщения  о  полупроводниковых тензорезистивных преобразователях механических величин,  

полученных с помощью технологии интегральной микроэлектроники, появились в печати в конце шестидесятых 
годов [1].  В начале семидесятых  годов интегральные  тензорезистивные  датчики  давления  (далее по тексту - 
ИТРДД) нашли широкое  применение  в  авиапромышленности [2], а в последствии с успехом перекочевали и в 
другие области народного хозяйства. 

1. Высокая технологичность производства. 
2. Сравнительно низкая стоимость. 
3. Малые размеры и высокое быстродействие. 
Это далеко неполный список  тех  достоинств,  благодаря которым  ИТРДД получили такое широкое 

использование за столь короткие сроки. 
Однако наряду  с перечисленными достоинствами ИТРДД обладают рядом недостатков: 
1.     Изменение  температуры  окружающей  среды приводит к сильному температурному дрейфу нуля и 

чувствительности измерительных мостов ИТРДД. 
2.     Имеет место сильный временной  дрейф,  обусловленный изменением  во  времени  номиналов 

резисторов измерительного моста ИТРДД. 
Однако если  во  втором  случае в основном используются различные методы старения датчиков (отжиг,  

электротермотренировка и т.д.),  то в первом случае нашли широкое использование различные схемотехнические 
методы компенсации температурного дрейфа.  Одному из таких способов компенсации температурного дрейфа 

нуля моста ИТРДД и посвящена данная статья. 
Использование мостовой  измерительной  схемы  (рис.1) 

позволяет улучшить параметры ИТРДД,  т.к.  в этом случае 
температурный дрейф нуля определяется не абсолютным значением 
ТКС,  входящих в нее терморезисторов, а их разбросом. Практически 
разброс ТКС может превышать 10%,  что приводит к снижению 
эффективности мостовой схемы и требует использования 
дополнительных мер [1]. 

Схемы термокомпенсации  нуля  по  характеру воздействия 
подразделяются на активные и пассивные. 

Активные схемы  термокомпенсации  температурного  дрейфа 
нуля можно разделить на три различные группы  в  зависимости от 
принципа работы [1]. 

К первой  группе относятся схемы непосредственно влияющие 
на напряжения и токи моста. 

Ко второй группе относяся схемы,  в которых компенсация 
Рис.1.  Схема моста сопротивлений.                 температурного дрейфа производится на этапе последующей  

                                                                    обработки в зависимости от температуры кристалла ИТРДД. 
К третьей группе относятся различные схемы, позволяющие поддерживать температуру ИТРДД на одном 

уровне, чем и устраняется температурный дрейф. 



Активные схемы  термокомпенсации  позволяют практически полностью компенсировать температурный 
дрейф моста, но большие аппаратные  затраты,  связанные  с реализацией этих схем компенсации делают порой 
более предпочтительным  использование пассивных схем термокомпенсации. 

В простейшем случае,  для случая линейной аппроксимации температурных  зависимостей,  условия баланса 
мостовой схемы можно описать системой из двух уравнений.  Первое  уравнение соответсвует начальной 
температурной точке,  а второе конечной. 

В начальной  температурной  точке условие баланса моста описывается уравнением: 
R1*R4 = R2*R3                                                                                  (1) 

где: 
       R1,R2,R3,R4 - резисторы моста (рис.1) 
Для конечной температурной точки условие баланса  запишется в следующем виде: 

N1+N4+N1*N4 = N2+N3+N2*N3                                                                  (2) 
где: 
       N1,N2,N3,N4 - ТКС соответствующих резисторов 
       N1=dR1/R1,N2=dR2/R2,N3=dR3/R3,N4=dR4/R4 
       dR1,dR2,dR3,dR4 - абсолютное изменение номиналов резисторов моста в выбранном температурном  

диапазоне. 
В литературе [1] и [2] условие (2) имеет следующий вид: 

N1+N4 = N2+N3. 
Выражение (2) можно записать в более простой форме: 

                       B1*B4 = B2*B3,                                                                            (3) 
если ТКС резисторов моста оценивать посредством коэффициента B, который равен: 

            B1=(R1+dR1)/R1=1+N1, B2=(R2+dR2)/R2=1+N2,                                                (4) 
B3=(R3+dR3)/R3=1+N3, B4=(R4+dR4)/R4=1+N4. 

Известны схемы термокомпенсации нуля  с  использованием термистора и без использования термистора. 
Наибольший интерес представляет схема компенсации  нуля с помощью шунтирующего (Rш) и добавочного 

(Rд) термостабильных резисторов,  без использования термистора (один из вариантов включения Rш, Rд показан на 
рис.2). 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

Рис. 2.  Схема температурной компенсации                                           Рис. 3.  Схема моста ИТРДД, позволяющая 
дрейфа нуля  с использованием внешних                                               включать термокомпенсирующие резисторы. 
термостабильных   резисторов  

 
Подобные схемы позволяют снизить аддитивную  составляющую температурной  погрешности более чем в 6 

раз [3].  В полупроводниковых датчиках фирмы WELWYN ELECTRIC [4] с помощью двух внешних  резисторов  
Rд  и  Rш осуществлена компенсация температурного дрейфа    нуля схемы,   который    составляет 0.01%/град. в 
диапазоне температур 0-60°C. 

Всего возможно 16 различных  вариантов  включения  компенсирующих резисторов, но после несложного 
анализа их можно минимизировать до 4 и даже до двух вариантов включения  термокомпенсирующих резисторов.  



Исходя  из  этого  чаще  всего используются разомкнутые мостовые схемы,  имеющие пять внешних выводов (Рис. 
3.). 

Для правильного выбора величин резисторов Rд иRш необходимо знать величины сопротивления и ТКС 
каждого резистора тензомоста. При измерении  величин  резисторов  тензомоста  большое значение имеет 
правильность выбора измерительной схемы.  Для учета варисторных свойств полупроводниковых тензорезисторов 
и эффекта саморазогрева тензомоста питающим током, выбранная схема измерения должна имитировать  реальный  
режим  питания тензомоста. 

Для измерения величин резисторов тензомоста использовались упрощенная и полная измерительные схемы,  
которые удовлетворяют поставленным выше условиям. 

Упрощенная измерительная схема: 
Если принять допущение, что резисторы R1,R4 и R2,R3 попарно равны,  то  можно  использовать  

измерительную  схему, изображенную на рис. 4. Такая схема измерения используется в основном для испытания 
замкнутых тензомостов. 

Рис. 4.  Упрощенная  схема измерения величин 
резисторов тензомоста. 

 
Здесь напряжение U1 пропорционально току 

питания  моста (коэффициент пропорциональности равен 
величине измерительного резистора Rи1),  напряжение 
U2 равно  напряжению  питания тензомоста, а  
напряжение  U3  равно  напряжению  разбаланса моста. 

Величины резисторов  тензомоста  находим  по  
следующим формулам: 
          R1=R4=(U2+U3)*Rи1/U1                                  (5) 
          R2=R3=(U2-U3)*Rи1/U1 

К достоинствам данной измерительной схемы 
можно отнести минимум затрат,  связанных с 
измерениями, и высокую информативность выходных 
параметров (напряжение U3 несет полную информацию 
о начальном разбалансе и дрейфе нуля тензомоста). 

К недостаткам  данной измерительной схемы 
можно отнести возможность неверной оценки величин  

резисторов  тензомоста, из-за несоблюдения допущения о попарном равенстве резисторов R1,R4 и R2,R3,  что  
может привести в последующем к  неполной компенсации температурного дрейфа нуля тензомоста ИТРДД. 

 
Полная измерительная схема: 
Указанного, для предыдущего варианта, 

недостатка лишена схема измерения величин 
резисторов тензомоста,  изображенная на рис. 5. 

 
Рис. 5. Полная схема измерения величин 

резисторов тензомоста. 
 
 
Здесь напряжение  U1 пропорционально 

току питания моста (коэффициент 
пропорциональности равен величине 
измерительного резистора Rи1),  напряжение U5 
отображает соотношения токов в ветвях 
тензомоста.  Напряжения U2, U3, U4, из-за 
наличия измерительных резисторов Rи2 и Rи3, 
отображают несколько другие величины, нежели 
напряжение питания моста, напряжение 
разбаланса моста и напряжение положительной 
ветви моста. 



Токи, протекающие в ветвях тензомоста, и  величины  резисторов находим по следующим формулам: 
                Iм=U1*Rи1                                                                                 (6) 

I1=(Iм*Rи3-U5)/(Rи3+Rи2) 
I2=(Iм*Rи2-U5)/(Rи3+Rи2) 

R4=U4/(I2-U4/Ru4)-Ri3 
R3=(U2-U4)/I2 

R2=(U4-U3)/I1-Rи2 
R1=(U2-U4+U3)/I1 

Программа на БЕЙСИКе, позволяющая рассчитывать величины резисторов тензомоста: 
10 cls:dim R(4),B(4) 
40 print "***************************************** ********" 
50 print " Володин В. Я., E-mail:valvol@farlep.net" 
80 print " Программа расчета величин резисторов тензомоста." 
90 print "***************************************** ********" 
100 print " Схема измерения -  Полная " 
110 print " Величины измерительных резисторов:" 
120 print " Rи1=1000ом; Rи2=2000ом; Rи3=2000ом" 
130 defdbl I,U,R 
140 Ri1=1000:Ri2=2000:Ri3=2000 
150 print:print "Ввод исходных данных:" 
160 input "Введи U1(В)";U1 
170 input "Введи U2(В)";U2 
180 input "Введи U3(mВ)";U3 
190 U3=0.001*U3 
200 input "Введи U4(В)";U4 
210 input "Введи U5(mВ)";U5 
220 U5=0.001*U5 
230 Im=U1/Ri1 
240 I1=(Im*Ri3-U5)/(Ri3+Ri2) 
250 I2=(Im*Ri2-U5)/(Ri3+Ri2) 
260 R(4)=(U4-Ri3*I2)/I2 
270 R(3)=(U2-U4)/I2 
280 R(2)=(U4-U3-Ri2*I1)/I1 
290 R(1)=(U2-U4+U3)/I1 
300 Rm=(R(1)+R(2))*(R(3)+R(4))/(R(1)+R(2)+R(3)+R(4)) 
310 Um=0.001*Rm 
320 Uout=1000*Um*(R(4)/(R(3)+R(4))-R(2)/(R(1)+R(2))) 
330 print "Резисторы моста:" 
340 print tab(5);"R1(Ом)=";R(1) 
350 print tab(5);"R2(Ом)=";R(2) 
360 print tab(5);"R3(Ом)=";R(3) 
370 print tab(5);"R4(Ом)=";R(4) 
380 print tab(5);"Rm(Ом)=";Rm 
390 print 
400 print "Если мост замкнут и питается током Im=1mA, то:" 
410 print tab(5);"Um(В)=";Um 
420 print tab(5);"Uout(мВ)=";Uout 
430 input "Печатать (Y)";A$ 
440 if A$="N" or A$="n" then 560 
450 input "Текст >";TEXT$ 
460 lprint TEXT$ 
470 lprint "Резисторы моста:" 
480 lprint tab(5);"R1(Ом)=";R(1) 
490 lprint tab(5);"R2(Ом)=";R(2) 
500 lprint tab(5);"R3(Ом)=";R(3) 
510 lprint tab(5);"R4(Ом)=";R(4) 
520 lprint tab(5);"Rm(Ом)=";Rm 



530 lprint "Если мост замкнут и питается током Im=1mA, то:" 
540 lprint tab(5);"Um(В)=";Um 
550 lprint tab(5);"Uout(мВ)=";Uout 
560 end 
Величины четырех резисторов тензомоста  несут  достаточно полную информацию о свойствах тензомоста .  

Но для идентификации тензомостов удобнее пользоваться таким понятием, как, например, напряжение  разбаланса.  
В этом смысле хотелось бы иметь более независимый параметр тензомоста,  чем напряжение разбаланса, которое 
зависит от тока (или напряжения) питания тензомоста.  Если тензомост представить  четырехполюсником, то  
разбаланс моста можно выразить через коэффициент передачи четырехполюсника.   

Рис. 6. Эквивалентная схема тензомоста 
 
В этом случае  можно перейти  к  эквивалентной схеме (рис. 6),  которая собственно и будет являться  

эквивалентной схемой моста. 
Здесь резистор  Rэ1 характеризует входное сопротивление моста, резистор Rэ3 выходное сопротивление 

моста и  резистор Rэ2 коэффициент  передачи моста.  Так как на практике тензомост обычно работает на большое 
входное сопротивление  измерительного усилителя, то величиной резистора Rэ3 можно пренебречь. Это позволяет 
перейти к упрощенной эквивалентной схеме. 

Рис. 7. Упрощенная эквивалентная схема тензомоста. 
 
Величины резисторов Rэ1 и Rэ2, упрощенной эквивалентной схемы можно рассчитать по следующим 

формулам: 
           Rэ1=(R1*R3+R2*(2*R3+R4))/(R1+R2+R3+R4)                                                  (7) 

Rэ2=(R1*R4-R2*R3)/(R1+R2+R3+R4) 
Входное сопротивление моста будет равно: 

         Rм=Rэ1+Rэ2=((R1+R2)*(R3+R4))/(R1+R2+R3+R4)                                               (8) 
Коэффициент передачи, в случае питания моста напряжением: 

      Ku=Rэ2/(Rэ1+Rэ2)=(R1*R4-R2*R3)/((R1+R2)*(R3+R4))                                           (9) 
Коэффициент передачи, в случае питания моста током: 

       Ki=Rэ2=(R1*R4-R2*R3)/(R1+R2+R3+R4)   Ом                                                (10) 
В случае если Ku=Ki=0 (мост  сбалансирован),  выражения (9)  и  (10)  трансформируются  в условие 

начального баланса (1). 
При питании моста стабильным током,  величиной сопротивления резистора Rэ1 так же можно пренебречь,  

так как он  оказывается включенным  последовательно с бесконечно большим сопротивлением источника тока.   
А так как обычно,  в целях компенсации температурного дрейфа тензочуствительности, тензомост питают от 

источника тока,  то в этом случае можно перейти к еще более простой эквивалентной схеме тензомоста (рис. 8). 
В этом случае можно говорить о сопротивлении разбаланса и сопротивлении чувствительности тензомоста. 



 

Рис. 8.  Упрощенная  эквивалентная схема тензомоста при питании током 

Рис. 9. Варианты включения термокомпенсирующих резисто ров: 
а) - П1Д2, б) - П1Д4, в) - П3Д2, г) - П3Д4. 



Как уже  говорилось выше,  тензомост практически всегда имеет начальный небаланс,  величина которого 
изменяется  при изменении температуры.  Всего возможно четыре варианта соотношений начальных значений 
величин резисторов  тензомоста  и их ТКС:                

 R1*R4 > R2*R3, B1*B4 > B2*B3                                                            (11) 
R1*R4 > R2*R3, B1*B4 < B2*B3 
R1*R4 < R2*R3, B1*B4 > B2*B3 
R1*R4 < R2*R3, B1*B4 < B2*B3 

- где соответствующий коэффициент B находят из выражения (4). 
Практически, с помощью резисторов Rш и Rд,  можно получить большое количество  вариантов  начального  

небаланса  и температурного дрейфа  тензомоста,  но в данном отчете будет рассмотрен частный случай, имеющий 
практический интерес,  который  предусматривает  полную компенсацию небаланса в начале и в конце выбранного 
температурного диапазона. Этот случай описывается уравнениями (1) и (3): 

R1*R4 = R2*R3 
B1*B4 = B2*B3 

Выбранная схема тензомоста (рис.3) допускает четыре варианта включения термокомпенсирующих 
резисторов (рис.9). 

Практика показала,  что на выбор схемы включения влияет тот показатель  тензомоста  (начальный небаланс 
или температурный дрейф),  который преобладает. Поэтому оценочные выражения (11) можно переписать в 
следующем виде: 

                       R1*R4 > R2*R3                                                                          (12) 
R1*R4 < R2*R3 
B1*B4 > B2*B3 
B1*B4 < B2*B3 

Соответствующие им схемы включения  термокомпенсирующих резисторов указаны в таблице 1. 
Таблица 1. 

Преобладающий показатель Схема включения резисторов 
R1*R4 > R2*R3 
R1*R4 < R2*R3 
B1*B4 > B2*B3 
B1*B4 < B2*B3 

П1Д2 
П3Д4 
П1Д4 
П3Д2 

Преобладающий показатель определяется по критерию положительности величин термокомпенсирующих 
резисторов. 

Схемы включения П1Д2 и П1Д4 соответственно эквивалентны П3Д4 и П3Д2.  В последнем легко убедиться, 
если резисторы R1 и R2 соответственно поменять местами с резисторами R3 и R4. 

Исходя из  сказанного анализ схем включения резисторов упрощается ровно в два раза. 
Вариант П1Д4 (П3Д2): 
Этот вариант включения выбирается в случае если, 

B1*B4 > B2*B3. 
В этом случае величина резистора Rш находится по формуле: 

R3*R2*R1*B1*(1-B3*B2) 
Rш =--------------------------------------------------------------------   Ом                                    (13) 

R3*R2*B3*B2-B1*(R4*R1*(B4-1)+R3*R2) 
Из условия баланса моста в начальной температурной точке (1), находим величину Rд: 

            Rд = (R3*R2*(R1+Rш)/(R1*Rш))-R4  Ом                                                     (14) 
Вариант П1Д2 (П3Д4): 
Этот вариант включения выбирается в случае если, 

R1*R4 > R2*R3. 
В этом случае для определения величины резистора Rш необходимо решить квадраное уравнение вида: 

                     A*Rш*Rш+B*Rш+C = 0                                                                    (15) 
где: 

A=B3*(R1*R4-R2*R3)+B2*B3*R2*R3-B1*B4*R1*R4 
B=R1*(B3*R2*R3*(B1+1)*(B2-1)+B1*R1*R4*(B3-B4)) 

C=R2*R3*B1*B3*R1*R1*(B2-1) 
Из условия баланса моста в начальной температурной точке (1), находим величину Rд: 

                   Rд=(R4*R1*Rш)/(R3*(R1+Rш))-R2                                                         (16) 
Рассчитанные по формулам (13), (14), (15), (16) величины резисторов Rш и Rд в уточнениях не нуждаются. 
 



Программа на БЕЙСИКе, позволяющая рассчитывать величины термокомпенсирующих резисторов Rш и Rд: 
10 cls:dim R(8),B(4) 
40 print "***************************************** ********" 
50 print "Володин В. Я., E-mail:valvol@farlep.net" 
70 print "Программа расчета величин термокомпенсирующих" 
80 print "резисторов тензомоста." 
90 print "***************************************** ********" 
100 print " Схема измерения -  Полная " 
110 print " Величины измерительных резисторов:" 
120 print " Rи1=1000ом; Rи2=2000ом; Rи3=2000ом" 
130 defdbl A-C,K,I,U,R:defint J,P,D 
140 Ri1=1000:Ri2=2000:Ri3=2000 
150 print:print "Ввод исходных данных !!!" 
160 print "Для минимальной температуры:" 
170 for J=0 to 4 step 4 
180 input "Введи U1(В)";U1 
190 input "Введи U2(В)";U2 
200 input "Введи U3(mВ)";U3 
210 U3=0.001*U3 
220 input "Введи U4(В)";U4 
230 input "Введи U5(mВ)";U5 
240 U5=0.001*U5 
250 Im=U1/Ri1 
260 I1=(Im*Ri3-U5)/(Ri3+Ri2) 
270 I2=(Im*Ri2-U5)/(Ri3+Ri2) 
280 R(4+J)=(U4-Ri3*I2)/I2 
290 R(3+J)=(U2-U4)/I2 
300 R(2+J)=(U4-U3-Ri2*I1)/I1 
310 R(1+J)=(U2-U4+U3)/I1 
320 Rm=(R(1+J)+R(2+J))*(R(3+J)+R(4+J))/(R(1+J)+R(2+J)_ 
    +R(3+J)+R(4+J)) 
330 Um=0.001*Rm 
340 Uout=1000*Um*(R(4+J)/(R(3+J)+R(4+J))-R(2+J)/(R(1+J)_ 
    +R(2+J))) 
350 print "Резисторы моста:" 
360 print tab(5);"R1(Ом)=";R(1+J) 
370 print tab(5);"R2(Ом)=";R(2+J) 
380 print tab(5);"R3(Ом)=";R(3+J) 
390 print tab(5);"R4(Ом)=";R(4+J) 
400 print tab(5);"Rm(Ом)=";Rm 
410 print 
420 print "Если мост замкнут и питается током Im=1mA, то:" 
430 print tab(5);"Um(В)=";Um 
440 print tab(5);"Uout(мВ)=";Uout 
450 print:print "Для максимальной температуры:" 
460 next J 
470 B(1)=1+(R(5)-R(1))/R(1) 
480 B(2)=1+(R(6)-R(2))/R(2) 
490 B(3)=1+(R(7)-R(3))/R(3) 
500 B(4)=1+(R(8)-R(4))/R(4) 
510 if B(1)*B(4)>B(2)*B(3) then P=1:D=4:PP=3:PD=2:goto 530 
520 P=3:D=2:PP=1:PD=4 
530 CRsh=R(PP)*R(PD)*R(P)*B(P)*(1-B(PP)*B(PD)) 
540 ZRsh=R(PP)*R(PD)*B(PP)*B(PD)-B(P)*(R(D)*R(P)*(B(D)-1)_ 
    +R(PP)*R(PD)) 
550 Rsh=CRsh/ZRsh 
560 Rd=(R(PP)*R(PD)*(R(P)+Rsh)/(R(P)*Rsh))-R(D) 



570 if Rsh>=0 and Rd>=0 then 730 
580 if R(1)*R(4)>R(2)*R(3) then P=1:D=2:PP=3:PD=4:goto 600 
590 P=3:D=4:PP=1:PD=2 
600 A=B(PP)*(R(PD)*R(P)-R(PP)*R(D))+B(PP)*B(D)*R(PP)*R(D)_ 
    -B(PD)*B(P)*R(PD)*R(P) 
610 B=R(P)*(B(PP)*R(PP)*R(D)*(B(P)+1)*(B(D)-1)+B(P)*R(PD)_ 
    *R(P)*(B(PP)-B(PD))) 
620 C=R(PP)*R(D)*B(PP)*B(P)*R(P)*R(P)*(B(D)-1) 
630 ID=B*B-4*A*C 
640 if ID<0 then 860 
650 KID=sqr(ID) 
660 for J=1 to 2 
670 Rsh=(KID-B)/(2*A) 
680 Rd=(R(PD)*R(P)*Rsh)/(R(PP)*(R(P)+Rsh))-R(D) 
690 if Rsh>=0 and Rd>=0 then 730 
700 KID=-KID 
710 next J 
720 goto 860 
730 print "Шунт включается в ветвь номер >";P 
740 print "Добавочное сопротивление в ветвь номер >";D 
750 print "Расчетные значения:" 
760 if Rsh<1E7 then print using "Rшунт.=####.### kОм";_ 
 Rsh/1000 else print using "Rшунт.=#.#####^^^^ kOm";Rsh/1000) 
770 if Rd<1E4 then print using "Rдоб. =####.### Ом";Rd_ 
    else print using "Rдоп. =#.#####^^^^ Om";Rd 
780 R(P)=R(P)*Rsh/(R(P)+Rsh):R(P+4)=R(P+4)*Rsh/(R(P+4)+Rsh) 
790 R(D)=R(D)+Rd:R(D+4)=R(D+4)+Rd 
800 Rk1=(R(1)*R(4)-R(2)*R(3))/(R(1)+R(2)+R(3)+R(4)) 
810 Rk2=(R(5)*R(8)-R(6)*R(7))/(R(5)+R(6)+R(7)+R(8)) 
820 print "После компенсации (для Im=1mA):" 
830 print using "Небаланс в начале = -#.###^^^^ mV";Rk1 
840 print using "Небаланс в  конце = -#.###^^^^ mV";Rk2 
850 end 
860 print "Компенсация невозможна !!!":goto 760 

